# Thomas decomposition of differential systems and its implementation in Maple

Vladimir Gerdt[1] Markus Lange-Hegermann[2] Daniel Robertz[3]

[1]Laboratory of Information Technologies
Joint Institute for Nuclear Research
141980 Dubna, Russia

[2]University of Applied Sciences Ostwestfalen-Lippe
32656 Lemgo, Germany

[3]School of Computing, Electronics and Mathematics
Plymouth University
PL4 8AA Plymouth, United Kingdom

PCA '2018, April 16, St.Petersburg, Russia

# Contents

# Symbolic Analysis of Nonlinear Differential Systems

**Goal:** Given a system of equations (and possibly inequations), extract from them as much information on its solutions as possible without (generally impossible) explicit integration/solving and/or "simplify / rewrite" the equations for the further numerical solving.

- Consistency
- Dimension of the solution space (arbitrariness in a general analytic solution of DEs)
- Elimination of a subset of variables
- Reduction to a finite subset of "smaller" problems with disjoint solution set
- Well posing of initial value problem (for PDEs)
- Computation of hidden constraints for dependent variables of DEs
- Rewriting into a triangular form
- Check whether an equation is valid for all common solutions of a system of equations
- ..........................................................................

**Universal algorithmic method (tool): Thomas decomposition**

# Input differential system

The fully algorithmic Thomas decomposition is applicable to a set of finite-order partial differential equations (PDEs) of the form

$$p_i\left(x_1,\ldots,x_n;u_1,\ldots,u_m,\ldots,\frac{\partial^{j_1+\cdots+j_n}u_k}{\partial x_1^{j_1}\cdots\partial x_n^{j_n}},\ldots\right)=0\,,\quad i=1,\ldots,s\,,$$

where $k=1,\ldots,m$ and $u_k=u_k(x_1,\ldots,x_n)$. It is assumed that $p_i$ are polynomials in their arguments.

The decomposition also allows enlargement of PDEs with a set of inequations

$$q_i\left(x_1,\ldots,x_n;u_1,\ldots,u_m,\ldots,\frac{\partial^{j_1+\cdots+j_n}u_k}{\partial x_1^{j_1}\cdots\partial x_n^{j_n}},\ldots\right)\neq 0\,,\quad i=1,\ldots,t\,,$$

where $q_k$ are also polynomials in their variables.

# (Cauchy-)Kovalevskaya theorem (1875)

## Theorem

*Let polynomials $p_i$ read*

$$p_i \;=\; \frac{\partial^{n_i} u_i}{\partial x_1^{n_i}} - F_i\left(x_1,\ldots,x_n; u_1,\ldots,u_m,\ldots,\frac{\partial^{j_1+\cdots+j_n} u_k}{\partial x_1^{j_1}\cdots\partial x_n^{j_n}},\ldots\right)$$

*where $j_1+\cdots+j_n = j \le n_i$ , $j_1 < n_i$ and all the functions $F_i$ (not necessarily polynomial) are analytic in a neighborhood of the point*

$$x_i = x_i^0,\;\; u_k = u_k^0,\;\; r^0_{k;\,j_1,\ldots,j_n} := \left.\frac{\partial^{j_1+\cdots+j_n} u_k}{\partial x_1^{j_1}\cdots\partial x_n^{j_n}}\right|_{x_1=x_1^0,\ldots,x_n=x_n^0} \qquad (i,k=1,\ldots,m=s)\,.$$

*Then in some neighborhood of the point $(x_1^0,\ldots,x_n^0)$ the PDE system $p_i=0$ $(i=1,\ldots,s)$ has a unique analytic solution satisfying the initial conditions*

$$\left\{\begin{array}{l} u_k = \phi_k(x_2,x_3,\ldots,x_n)\,, \\ \frac{\partial u_k}{\partial x_1} = \phi_{k;\,1}(x_2,x_3,\ldots,x_n)\,, \\ \ldots\ldots\ldots\ldots\ldots\ldots\ldots\ldots\ldots\ldots\ldots\ldots \\ \frac{\partial^{n_k-1} u_k}{\partial x_1^{n_k-1}} = \phi_{k;\,n_k-1}(x_2,x_3,\ldots,x_n)\,, \end{array}\right. \qquad \textit{for } x_1 = x_1^0\,,\;\; k=1,\ldots,m\,,$$

*where all $\phi$ are arbitrary analytic functions of their arguments in a neighborhood of the point $(x_2^0,\ldots,x_n^0)$ such that they take at this point the initial values.*

# Generalizations

- E.Cartan (1899,1901) - exterior differential systems generated by 1-forms (Pfaffian systems) and notion of involution (2004)
- Riquier-Janet theory:

  Ch.Riquier (1910) - orthonomic (quasilinear) and passive PDE systems

  M.Janet (1920,1929) - theory of Janet monomials
- E.Kähler (1934) - extension of the Cartan theory to any differential ideal generated by an exterior differential system.
- J.Thomas (1937) - disjoint decomposition of polynomially nonlinear PDE systems into passive subsystems
- V.Gerdt and Yu.Blinkov (1998) - theory of involutive monomial divisions and involutive bases
- T.Bächler,V.Gerdt,M.Lange-Hegermann and D.Robertz (2012) algorithmization of algebraic and differential Thomas decomposition and implementation of the last decomposition in Maple (M.Lange-Hegermann).

# Differential polynomial ring

The set of all expressions potentially occurring as $p_i$ and $q_j$ is the smallest polynomial ring containing all partial derivatives of $u_1$, $u_2$, ..., $u_m$, namely the differential polynomial ring

$$R := \mathbb{Q}\{u_1, \ldots, u_m\} := \mathbb{Q}[\,(u_k)_J \mid k \in \{1, \ldots, m\},\, J \in (\mathbb{Z}_{\geq 0})^n\,].$$

Here the differential indeterminate $u_k = (u_k)_{(0,\ldots,0)}$ represents the unknown function $u_k(x_1, \ldots, x_n)$ with the same name and, more generally, $(u_k)_J$ for the multi-index $J = (j_1, j_2, \ldots, j_n) \in (\mathbb{Z}_{\geq 0})^n$ represents the partial derivative

$$\frac{\partial^{j_1+j_2+\ldots+j_n} u_k}{\partial x_1^{j_1} \partial x_2^{j_2} \ldots \partial x_n^{j_n}}.$$

The ring $R$ is closed under the derivations $\partial_1$, ..., $\partial_n$ acting as

$$\partial_i\,(u_k)_J := (u_k)_{J+1_i}, \qquad J + 1_i := (j_1, \ldots, j_{i-1}, j_i + 1, j_{i+1}, \ldots, j_n).$$

The coefficient field $\mathbb{Q}$ of rational numbers can also be replaced with a larger field containing $\mathbb{Q}$ admitting $n$ derivations of which then the derivations $\partial_i$ are extensions to $R$. For example, the coefficient field can be chosen to be the field $\mathbb{Q}(x_1, \ldots, x_n)$ of rational functions in $x_1$, ..., $x_n$ with the usual derivations, if the system to be dealt with consists of PDEs with rational function coefficients.

# Differential ideal

Given a system of partial differential equations

$$p_1 = 0\,, \quad p_2 = 0\,, \quad \ldots, \quad p_s = 0\,, \quad p_i \in R\,,$$

the minimal subset of $R$ which contains $P := \{p_i \mid i = 1, \ldots, s\}$ and is closed under taking linear combinations of its elements with coefficients in $R$ and under differentiation is called differential ideal of $R$ generated by $P$. A differential ideal of $R$ which contains for each of its elements $p$ also all differential polynomials in $R$ of which a power is equal to $p$ is said to be radical.

The differential ideal of all $p \in R$ which vanish under substitution of any analytic solution of a PDE system is a radical differential ideal. The following important theorem establishes a one-to-one correspondence between radical differential ideals of $R$ and solutions sets (with complex analytic functions on suitable domains) of PDE systems which are defined over $R$.

## Theorem (Nullstellensatz of Ritt-Raudenbush)

*Let $I$ be the differential ideal of $R$ generated by the left hand sides $p_1, p_2, \ldots, p_s$ of a PDE system. If a differential polynomial $p \in R$ vanishes under substitution of any analytic solution of $p_1 = 0, \ldots, p_s = 0$, then some power of $p$ is an element of $I$.*

Radical differential ideals are finitely generated in the following sense.

## Theorem (Basis Theorem of Ritt-Raudenbush)

*For every radical differential ideal $I$ of $R$ there exists a finite subset $B$ of $I$ such that $I$ is the smallest radical differential ideal of $R$ which contains $B$.*

# Ranking

A ranking $\succ$ on $R$ is a total ordering on

$$\Theta u \;:=\; \{\, (u_k)_J \mid k \in \{1,\ldots,m\},\, J \in (\mathbb{Z}_{\geq 0})^n \,\}$$

such that the following two conditions are satisfied.

1. For all $i \in \{1,\ldots,n\}$ and $k \in \{1,\ldots,m\}$ we have $\partial_i\, u_k \succ u_k$.
2. For all $i \in \{1,\ldots,n\}$, $k, l \in \{1,\ldots,m\}$, $K, L \in (\mathbb{Z}_{\geq 0})^n$, the implication $(u_k)_K \succ (u_l)_L \;\Rightarrow\; \partial_i\,(u_k)_K \succ \partial_i\,(u_l)_L$ holds.

Of special interest to us is a Riquier ranking such that

$$\forall\, \delta_1, \delta_2 \in \Theta := \left\{ \frac{\partial^{j_1+\cdots+j_n}}{\partial x_1^{j_1} \cdots \partial x_n^{j_n}} \;\middle|\; j_1, \ldots, j_n \in \mathbb{Z}_{\geq 0} \right\}$$

we have $\delta_1 u \succ \delta_2 u$ if the total differentiation order in $\delta_1 u$ is greater than the one in $\delta_2 u$ for any dependent variable $u$, and the following condition holds for all dependent variables $v$, $w$:

$$\delta_1 v \succ \delta_2 v \quad\Longrightarrow\quad \delta_1 w \succ \delta_2 w\,.$$

# Reduction I

Let a ranking $\succ$ on $R$ be fixed.

- Every $p \in R \setminus 0$ involves a symbol $(u_k)_J$ which is maximal with respect to $\succ$. is called leader of $p$ and denoted by $\mathrm{ld}(p)$ .
- The coefficient of the highest power of $\mathrm{ld}(p)$ in $p$ is called initial of $p$ and denoted by $\mathrm{init}(p)$.
- The formal derivative of $p$ with respect to $\mathrm{ld}(p)$ is called the separant of $p$ and denoted by $\mathrm{sep}(p)$.

Consider two non-constant polynomials $p_1$ and $p_2$ in $R$. If $\mathrm{ld}(p_1) = \mathrm{ld}(p_2)$ we consider the degrees $d_1$ and $d_2$ of $p_1$ and $p_2$ in $v := \mathrm{ld}(p_1)$, respectively. If $d_1 \geq d_2$, then

$$\mathrm{init}(p_2)\, p_1 - \mathrm{init}(p_1)\, v^{d_1 - d_2}\, p_2$$

is either constant, or has a leader which is ranked lower than $v$, or has the same leader, but has smaller degree in $\mathrm{ld}(p_1)$ than $p_1$.
If $d_1 < d_2$, no reduction of $p_1$ modulo $p_2$ is possible.

# Reduction II

If $\mathrm{ld}(p_1) \neq \mathrm{ld}(p_2)$, but there exists $J \in (\mathbb{Z}_{\geq 0})^n$ such that $v := \mathrm{ld}(p_1) = \partial^J \mathrm{ld}(p_2)$, then

$$\mathrm{sep}(p_2)\, p_1 - \mathrm{init}(p_1)\, v^{d_1 - 1}\, \partial^J p_2$$

is either constant or has a leader which is ranked lower than $v$ (because $\partial^J p_2$ has degree one in $v$.)

If no $J \in (\mathbb{Z}_{\geq 0})^n$ exists such that $\mathrm{ld}(p_1) = \partial^J \mathrm{ld}(p_2)$, then no reduction of $p_1$ modulo $p_2$ is possible.

This reduction process can be adapted so as to eliminate any occurrence (in sufficiently high degree) of symbols $(u_k)_J$ which are leaders or derivatives of leaders of $p_1, \ldots, p_s \in R$ in a given differential polynomial $p \in R$.

We say that $p \in R$ reduces to zero modulo $p_1, \ldots, p_s \in R$ and their derivatives if $p$ can be reduced to the zero polynomial in this way.

# Simple differential systems I

A system of partial differential equations and inequations

$$p_1 = 0\,,\ \ldots\,,\ p_s = 0\,, \quad q_1 \neq 0\,,\ \ldots\,,\ q_t \neq 0\,, \qquad (s, t \in \mathbb{Z}_{\geq 0}) \qquad (1)$$

where $p_1, \ldots, p_s, q_1, \ldots, q_t$ are non-constant differential polynomials in $R$, is said to be simple (with respect to $\succ$) if the following conditions are satisfied.

1. The leaders of $p_1, p_2, \ldots, p_s, q_1, q_2, \ldots, q_t$ are pairwise different (triangularity).
2. Let $v_1 \succ v_2 \succ \ldots \succ v_k$ be the elements of $\Theta u$ which effectively occur in the differential polynomials $p_1, \ldots, p_s, q_1, \ldots, q_t$. We consider (1) as a system of polynomial equations and inequations in $v_1, v_2, \ldots, v_k$.

# Simple differential systems II

If $r \in \{p_1, \ldots, p_s, q_1, \ldots, q_t\}$ has leader $v_\ell$, thus is a polynomial $r(v_\ell, v_{\ell+1}, \ldots, v_k)$, then we require that for every solution $(a_1, a_2, \ldots, a_k) \in \mathbb{C}^k$ of (1) the polynomial $r(v_\ell, a_{\ell+1}, a_{\ell+2}, \ldots, a_k)$ has the same degree in $v_\ell$ as $r(v_\ell, v_{\ell+1}, \ldots, v_k)$ and has no multiple roots.
Equivalently, the initial of $r$ and the discriminant of $r$ with respect to $v_\ell$ do not vanish on the solution set of (1) in $\mathbb{C}^k$.

3. The differential consequences of $p_1 = 0$, $p_2 = 0$, ..., $p_s = 0$ contain all integrability conditions of this PDE system, i.e., the cross-derivative of each pair of distinct equations whose leaders involve the same unknown function reduces to zero modulo $p_1$, ..., $p_s$ and their derivatives (passivity / involutivity or formal integrability).
4. No reduction of $q_1$, $q_2$, ..., $q_t$ is possible modulo $p_1$, $p_2$, ..., $p_s$ and their derivatives.

# Thomas decomposition

Let $S$ be a system of partial differential equations and inequations, defined over $R$.

A Thomas decomposition of $S$ (with respect to $\succ$) is a finite collection of simple differential systems $S_1, \ldots, S_r$, defined over $R$, such that the solution sets of $S_1, \ldots, S_r$ form a partition of the solution set of $S$.

The method outlined above allows to compute a Thomas decomposition for any differential system $S$ as considered above, with respect to any ranking $\succ$, in finitely many steps (Baechler, Gerdt, Lange-Hegermann, Robertz'12).

However, a Thomas decomposition of a differential system is not uniquely determined in general. The relevance of simple differential systems and the decomposition of a general differential system into simple differential systems is explained by the following theorem.

# Characterization of vanishing ideal I

## Theorem (Robertz'14)

*Let a simple differential system $S$ be given by*

$$p_1 = 0\,, \quad \ldots\,, \quad p_s = 0\,, \quad q_1 \neq 0\,, \quad \ldots\,, \quad q_t \neq 0\,, \qquad (s, t \in \mathbb{Z}_{\geq 0})$$

*where $p_1, \ldots, p_s, q_1, \ldots, q_t \in R$. Let $E$ be the differential ideal of $R$ which is generated by $p_1, p_2, \ldots, p_s$. Moreover, let $q \in R$ be the product of the initials and separants of $p_1, p_2, \ldots, p_s$. Then the differential ideal*

$$E : q^{\infty} \;:=\; \{\, p \in R \mid q^r\, p \in E \text{ for some } r \in \mathbb{Z}_{\geq 0} \,\}$$

*is equal to the set of differential polynomials in $R$ which vanish under substitution of any analytic solution of $S$. In particular, it is a radical differential ideal. A differential polynomial $p$ is an element of $E : q^{\infty}$ if and only if $p$ reduces to zero modulo $p_1, \ldots, p_s$ and their derivatives.*

# Characterization of vanishing ideal II

## Corollary

*Let $S$ be a simple differential system as in the above theorem. For each $k \in \{1, \ldots, m\}$ let $\partial^{J_{k,1}}, \partial^{J_{k,2}}, \ldots, \partial^{J_{k,n_k}}$, where $J_{k,i} \in (\mathbb{Z}_{\geq 0})^n$, $n_k \in \mathbb{Z}_{\geq 0}$, be the differential operators such that $\partial^{J_{k,i}} u_k = \mathrm{ld}(p_{j_i})$ for some $j_i \in \{1, \ldots, s\}$. Due to the characterization of the vanishing ideal for $S$, the set of principal derivatives*

$$P := \bigcup_{k=1}^{m} \bigcup_{i=1}^{n_k} \{ \partial^J \partial^{J_{k,i}} u_k \mid J \in (\mathbb{Z}_{\geq 0})^n \}$$

*consists of those elements $v \in \Theta u$ for which there exists an equation with leader $v$ that is a consequence of $S$. We refer to the elements of the complement $\Theta u \setminus P$ as the parametric derivatives. Note that $\mathrm{ld}(q_j) \in \Theta u \setminus P$ for all $j = 1, \ldots, t$ because of the properties of simple systems.*

# Implementation

- The differential Thomas decomposition algorithm implemented as the Maple package TDDS (Thomas Decomposition of Differential Systems)
- Programming language: Maple 11 to Maple 2018
- Downloadable from: https://wwwb.math.rwth-aachen.de/thomasdecomposition/index.php
- Licensing provisions: GNU LPGL license
- Included as a library module in Maple 2018
- The built-in algorithm is based on pseudo-division of differential polynomials, as in Euclid's algorithm, with case distinctions according to vanishing or non-vanishing leading coefficients and discriminants, combined with completion to involution for PDEs. Since an enormous growth of expressions can be expected in general, efficient versions of these techniques need to be used, e.g., subresultants, Janet division, and need to be applied in an appropriate order. In addtion, factorization of polynomials, while not strictly necessary for the method, should be utilized to reduce the size of expressions whenever possible.

# Consistency check

```
> restart;
> with(DifferentialThomas):
> ComputeRanking([x,y],[u]):
> Eq1:=[u[0,1]*u[1,0]+u[1,0]+1,u[2,0]*u[0,0]-u[0,1]^2+u[0,0]];
```

$$Eq1 := [u_{0,1}\,u_{1,0} + u_{1,0} + 1,\ u_{0,0}\,u_{2,0} - u_{0,1}^2 + u_{0,0}] \tag{1.1}$$

```
> TD:=DifferentialThomasDecomposition(Eq1,[]);
```

$$TD := [\,] \tag{1.2}$$

This shows that system Eq1 is inconsistent.

```
> Eq2:=[u[0,1]*u[1,0]+u[1,0]+1,u[2,0]*u[0,0]-u[0,1]^2-u[1,0]];
```

$$Eq2 := [u_{0,1}\,u_{1,0} + u_{1,0} + 1,\ u_{0,0}\,u_{2,0} - u_{0,1}^2 - u_{1,0}] \tag{1.3}$$

```
> TD:=DifferentialThomasDecomposition(Eq2,[]);
```

$$TD := [\mathit{DifferentialSystem}] \tag{1.4}$$

This shows that system Eq2 is consistent.

```
> map(print@JetList2Diff,DifferentialSystemEquations(TD[1]));
```

$$\left(\frac{\partial}{\partial x}u(x,y)\right)\left(\frac{\partial}{\partial y}u(x,y)\right) + \frac{\partial}{\partial x}u(x,y) + 1$$

$$\left(\frac{\partial}{\partial y}u(x,y)\right)^3 + \left(\frac{\partial}{\partial y}u(x,y)\right)^2 - 1$$

$$[\,] \tag{1.5}$$

Namely, there is a non-empty set of equations in the output simple system.

```
> Eq3:=[u[0,1]*u[1,0]+u[1,0]+1,u[2,0]*u[0,0]-u[0,1]^2-u[1,0]+a[0,0]*u[0,0],a[1,0],a[0,1]];
```

$$Eq3 := [u_{0,1}\,u_{1,0} + u_{1,0} + 1,\ a_{0,0}\,u_{0,0} + u_{0,0}\,u_{2,0} - u_{0,1}^2 - u_{1,0},\ a_{1,0},\ a_{0,1}] \tag{1.6}$$

Eq3 is the extension of Eq2 with the constant (parameter) 'a'.

```
> ComputeRanking([x,y],[[u],[a]]):
> TD:=DifferentialThomasDecomposition(Eq3,[]);
```

$$TD := [\mathit{DifferentialSystem}] \tag{1.7}$$

```
> op(PrettyPrintDifferentialSystem(TD[1]));
```

$$-\left(\frac{\partial}{\partial x}u(x,y)\right)^3 - \left(\frac{\partial}{\partial x}u(x,y)\right)^2 - 2\left(\frac{\partial}{\partial x}u(x,y)\right) - 1 = 0,\ \left(\frac{\partial}{\partial y}u(x,y)\right)\left(\frac{\partial}{\partial x}u(x,y)\right) + \frac{\partial}{\partial x}u(x,y) + 1 = 0,\ a(x,y) = 0,\ u(x,y) \neq 0 \tag{1.8}$$

The last equation in the output system shows that consistency holds if and only if a = 0.

## Computation of Lagrangian constraints (Eq.8.1, A. Deriglazov. Classical mechanics, Hamiltonian and Lagrangian formalism. Springer, Heidelberg, 2010)

```
> restart;
> with(DifferentialThomas):
```

The below application of the differential Thomas Decomposition for this problem taken from V.P.Gerdt, D.Robertz, Lagrangian constraints and differential Thomas decomposition, Advances in Applied Mathematics, 72, 113-138, 2016

```
> ivar:=[t];dvar:=[q1,q2];
```

$$ivar := [t]$$

$$dvar := [q1, q2] \quad (3.1)$$

```
> ComputeRanking(ivar,dvar);
```

Construction of an appropriate ranking

$$var := [q1_1, q2_1, q1_0, q2_0] \quad (3.2)$$

```
> L:=q2[0]^2*q1[1]^2+q1[0]^2*q2[1]^2+2*q1[0]*q2[0]*q1[1]*q2[1]+q1[0]^2+q2[0]^2; # Lagrangian
  EL:=map(a->PartialDerivative(diff(L,a[1]),t)-diff(L,a[0]),dvar); # Euler-Lagrange equations
```

$$L := q1_0^2 q2_1^2 + 2\, q1_0\, q2_0\, q1_1\, q2_1 + q2_0^2\, q1_1^2 + q1_0^2 + q2_0^2$$

$$EL := \left[2\, q1_0\, q2_0\, q2_2 + 4\, q1_1\, q2_0\, q2_1 + 2\, q1_2\, q2_0^2 - 2\, q1_0,\ 2\, q1_0^2\, q2_2 + 4\, q1_1\, q1_0\, q2_1 + 2\, q1_0\, q1_2\, q2_0 - 2\, q2_0\right] \quad (3.3)$$

```
> TD:=DifferentialThomasDecomposition(EL,[]);
```

$$TD := [DifferentialSystem, DifferentialSystem, DifferentialSystem] \quad (3.4)$$

```
> PrettyPrintDifferentialSystem(TD[1]);
```

$$\left[q1(t) + q2(t) = 0,\ 2\, q2(t) \left(\frac{d^2}{dt^2}\, q2(t)\right) + 2 \left(\frac{d}{dt}\, q2(t)\right)^2 - 1 = 0,\ q2(t) \neq 0\right] \quad (3.5)$$

the first expression is the local constraint q1(t)+q2(t)=0

```
> PrettyPrintDifferentialSystem(TD[2]);
```

$$\left[q1(t) - q2(t) = 0,\ 2\, q2(t) \left(\frac{d^2}{dt^2}\, q2(t)\right) + 2 \left(\frac{d}{dt}\, q2(t)\right)^2 - 1 = 0,\ q2(t) \neq 0\right] \quad (3.6)$$

the first expression is the local constraint q1(t)-q2(t)=0

```
> PrettyPrintDifferentialSystem(TD[3]);
```

$$[q1(t) = 0,\ q2(t) = 0] \quad (3.7)$$

the complementary constraints complementary to those in (3.5) and (3.6)

## Cole-Hopf transformation (Ex.3.8, T. Baechler, V. Gerdt, M. Lange-Hegermann, D. Robertz. Algebraic Thomas decomposition of algebraic and differential systems. Journal of Symbolic Computation, 47, 1233-1266, 2012)

We demonstrate how to study the Cole-Hopf transformation by using the differential Thomas decomposition.

```
> restart;
```

The claim is that for every non-zero analytic solution of the heat equation

```
> eta[1,0]+eta[0,2]=0;
```

$$\eta_{1,0} + \eta_{0,2} = 0 \tag{7.1}$$

the function defined by

```
> zeta=eta[0,1]/eta[0,0];
```

$$\zeta = \frac{\eta_{0,1}}{\eta_{0,0}} \tag{7.2}$$

is a solution to Burgers' equation

```
> zeta[1,0]+zeta[0,2]+2*zeta[0,1]*zeta[0,0]=0;
```

$$2\,\zeta_{0,1}\,\zeta_{0,0} + \zeta_{0,2} + \zeta_{1,0} = 0 \tag{7.3}$$

```
> with(DifferentialThomas):
```

We define a ranking on the ring of differential polynomials in eta and zeta such that any partial derivative of eta is ranked higher than any partial derivative of zeta.

```
> ComputeRanking([t,x], [[eta],[zeta]]):
```

We define the differential system which combines the heat equation in eta and Burgers' equation in zeta:

```
> CH := [eta[1,0]+eta[0,2], eta[0,0]*zeta[0,0]-eta[0,1]];
```

$$CH := \left[\eta_{1,0} + \eta_{0,2},\ \eta_{0,0}\,\zeta_{0,0} - \eta_{0,1}\right] \tag{7.4}$$

```
> map(print@JetList2Diff,CH):
```

$$\frac{\partial}{\partial t}\,\eta(t,x) + \frac{\partial^2}{\partial x^2}\,\eta(t,x)$$

$$\eta(t,x)\,\zeta(t,x) - \left(\frac{\partial}{\partial x}\,\eta(t,x)\right) \tag{7.5}$$

We also include the assumption eta <> 0 as an inequation.

```
> TD := DifferentialThomasDecomposition(CH, [eta]);
```

$$TD := [\mathit{DifferentialSystem}] \tag{7.6}$$

```
> PrettyPrintDifferentialSystem(TD[1]);
```

$$\left[\eta(t,x)\,\zeta(t,x)^2 + \eta(t,x)\left(\frac{\partial}{\partial x}\,\zeta(t,x)\right) + \frac{\partial}{\partial t}\,\eta(t,x) = 0,\ \eta(t,x)\,\zeta(t,x) - \left(\frac{\partial}{\partial x}\,\eta(t,x)\right) = 0,\ 2\left(\frac{\partial}{\partial x}\,\zeta(t,x)\right)\zeta(t,x) + \frac{\partial^2}{\partial x^2}\,\zeta(t,x) + \frac{\partial}{\partial t}\,\zeta(t,x) = 0,\ \eta(t,x) \neq 0\right] \tag{7.7}$$

The simple system of the resulting Thomas decomposition allows to read off that zeta as defined above is a solution of Burgers' equation if eta is a solution of the heat equation, which proves the original claim. Conversely, since the above simple differential system is consistent with the heat equation for eta by construction, we conclude that for any solution zeta of Burgers' equation there exists a solution eta of the heat equation such that the Cole-Hopf transformation of eta is zeta.

## Singular solution of ODE

**Ex.2.2.60, D. Robertz. Formal Algorithmic Elimination for PDEs. Lecture Notes in Mathematics, Vol. 2121. Springer, Cham, 2014**

```
> restart;
> with(DifferentialThomas):
>
```

This example demonstrate that the differential Thomas decomposition naturally distinguishes cases so that singular solutions are separated from the general solution.

```
> ivar := [t];
```

$$ivar := [t] \tag{9.3}$$

```
> dvar := [u];
```

$$dvar := [u] \tag{9.4}$$

```
> ComputeRanking(ivar, dvar);
```

We consider the following nonlinear ODE:

```
> L := [diff(u(t),t)^2-4*t*diff(u(t),t)-4*u(t)+8*t^2];
```

$$L := \left[\left(\frac{d}{dt}\,u(t)\right)^2 - 4\,t\left(\frac{d}{dt}\,u(t)\right) - 4\,u(t) + 8\,t^2\right] \tag{9.5}$$

```
> TD := DifferentialThomasDecomposition(L, []);
```

$$TD := [DifferentialSystem, DifferentialSystem] \tag{9.6}$$

The first simple system of the Thomas decomposition yields the general solution of the given ODE:

```
> S1 := PrettyPrintDifferentialSystem(TD[1]);
```

$$S1 := \left[\left(\frac{d}{dt}\,u(t)\right)^2 - 4\,t\left(\frac{d}{dt}\,u(t)\right) - 4\,u(t) + 8\,t^2 = 0, t^2 - u(t) \neq 0\right] \tag{9.7}$$

The inequation contained in the first simple system is a consequence of the assumption that the separant of the given ODE does not vanish.

The second simple system defines a singular solution, which is a solution of the given ODE for which the separant vanishes.

```
> S2 := PrettyPrintDifferentialSystem(TD[2]);
```

$$S2 := [t^2 - u(t) = 0] \tag{9.8}$$

As illustration we plot a few trajectories belonging to the general solution given by S1 as well as the singular solution given by S2 in the same diagram.

```
> P := map(c->2*((t+c)^2+c^2), [$-5..10]);
```

$$P := [2\,(t-5)^2+50, 2\,(t-4)^2+32, 2\,(t-3)^2+18, 2\,(t-2)^2+8, 2\,(-1+t)^2+2, 2\,t^2, 2\,(t+1)^2+2, 2\,(t+2)^2+8, 2\,(t+3)^2+18, 2\,(t+4)^2+32, 2\,(t+5)^2+50, 2\,(t+6)^2+72, 2\,(t+7)^2+98, 2\,(t+8)^2+128, 2\,(t+9)^2+162, 2\,(t+10)^2+200] \tag{9.9}$$

```
> plot([t^2, op(P)], t=-30..20);
```




Note that the singular solution is an envelope of the general solution.

## An example of an ODE provided by E. Cheb-Terrab

This example demonstrates different behavior of the packages DifferentialThomas, DifferentialAlgebra and DEtools.

```
> restart;
> with(DifferentialThomas):
>
> ivar := [x];
```

$$ivar := [x] \tag{10.1}$$

```
> dvar := [y];
```

$$dvar := [y] \tag{10.2}$$

```
> ComputeRanking(ivar, dvar);
> L := [(2*y(x)*diff(y(x),x,x)-diff(y(x),x)^2)^3 + 32*diff(y(x),x,x)*(x*diff(y(x),x,x) - diff(y(x),x))^3];
```

$$L := \left[\left(2\,y(x)\left(\frac{d^2}{dx^2}\,y(x)\right)-\left(\frac{d}{dx}\,y(x)\right)^2\right)^3+32\left(\frac{d^2}{dx^2}\,y(x)\right)\left(x\left(\frac{d^2}{dx^2}\,y(x)\right)-\left(\frac{d}{dx}\,y(x)\right)\right)^3\right] \tag{10.3}$$

```
> LL := Diff2JetList(L);
```

$$LL := \left[\left(2\,y_0 y_2 - y_1^2\right)^3 + 32\,y_2\left(x\,y_2 - y_1\right)^3\right] \tag{10.4}$$

```
> T := DifferentialThomasDecomposition(LL, []);
```

$$T := [\mathit{DifferentialSystem}, \mathit{DifferentialSystem}, \mathit{DifferentialSystem}, \mathit{DifferentialSystem}, \mathit{DifferentialSystem}] \tag{10.5}$$

We obtain a Thomas decomposition for the given ODE with five simple differential systems.

```
> PrettyPrintDifferentialSystem(T[1]);
```

$$\begin{aligned}\Bigg[&32\,x^3\left(\frac{d^2}{dx^2}\,y(x)\right)^4-96\,x^2\left(\frac{d}{dx}\,y(x)\right)\left(\frac{d^2}{dx^2}\,y(x)\right)^3+8\,y(x)^3\left(\frac{d^2}{dx^2}\,y(x)\right)^3-12\,y(x)^2\left(\frac{d}{dx}\,y(x)\right)^2\left(\frac{d^2}{dx^2}\,y(x)\right)^2+6\,y(x)\left(\frac{d}{dx}\,y(x)\right)^4\left(\frac{d^2}{dx^2}\,y(x)\right)-\left(\frac{d}{dx}\,y(x)\right)^6\\&+96\,x\left(\frac{d}{dx}\,y(x)\right)^2\left(\frac{d^2}{dx^2}\,y(x)\right)^2-32\left(\frac{d}{dx}\,y(x)\right)^3\left(\frac{d^2}{dx^2}\,y(x)\right)=0, y(x)^2+8\,x\neq 0, -y(x)^2+8\,x\neq 0, x\left(\frac{d}{dx}\,y(x)\right)-2\,y(x)\neq 0, \frac{d}{dx}\,y(x)\neq 0, y(x)\neq 0,\\&512\,x^3\left(\frac{d}{dx}\,y(x)\right)^3-768\,x^2 y(x)\left(\frac{d}{dx}\,y(x)\right)^2+27\,y(x)^4\left(\frac{d}{dx}\,y(x)\right)-48\,x\,y(x)^2\left(\frac{d}{dx}\,y(x)\right)+1728\,x^2\left(\frac{d}{dx}\,y(x)\right)-64\,y(x)^3\neq 0\Bigg]\end{aligned} \tag{10.6}$$

```
> PrettyPrintDifferentialSystem(T[2]);
```

$$\left[512\,x^3\left(\frac{d}{dx}\,y(x)\right)^3-768\,x^2 y(x)\left(\frac{d}{dx}\,y(x)\right)^2+27\,y(x)^4\left(\frac{d}{dx}\,y(x)\right)-48\,x\,y(x)^2\left(\frac{d}{dx}\,y(x)\right)+1728\,x^2\left(\frac{d}{dx}\,y(x)\right)-64\,y(x)^3=0, y(x)^2+8\,x\neq 0, -y(x)^2+8\,x\neq 0\right] \tag{10.7}$$

```
> PrettyPrintDifferentialSystem(T[3]);
```

$$\left[x\left(\frac{d}{dx}\,y(x)\right)-2\,y(x)=0, y(x)\neq 0, y(x)^2+8\,x\neq 0\right] \tag{10.8}$$

```
> PrettyPrintDifferentialSystem(T[4]);
```

$$\left[\frac{d}{dx}\,y(x)=0, y(x)\neq 0\right] \tag{10.9}$$

```
> PrettyPrintDifferentialSystem(T[5]);
```

$$\left[-y(x)^2+8\,x=0\right] \tag{10.10}$$

The DifferentialAlgebra package and the command rifsimp (in Maple 2017) do not terminate on the same input in reasonable time.

```
> with(DifferentialAlgebra):
> R := DifferentialRing(blocks=[y], derivations=[x]);
```

$$R := \mathit{differential\_ring} \tag{10.11}$$

```
> RosenfeldGroebner(L, R);
Warning,  computation interrupted
> DEtools[rifsimp](L, casesplit);
Warning,  computation interrupted
>
```

# Conclusions

- Differential Thomas decomposition provides a universal algorithmic tool to study and solve systems of polynomially nonlinear PDEs which can be enlarged with polynomially nonlinear inequations.
- The decomposition algorithm outputs a finite set of simple (i.e. triangular, involutive and without multiple solutions) differential subsystems whose solutions partition the solution set of the input system.
- The implementation of the algorithm has been done in Maple. It is freely available and included in Maple 2018.
- For more details on the decomposition algorithm and its implementation we refer to the paper (and to the references therein):
  V. Gerdt, M. Lange-Hegermann, D. Robertz. The MAPLE package TDDS for computing Thomas decompositions of systems of nonlinear PDEs. arXiv:1801.09942 [physics.comp-ph]